

# Excel SUMIF function - syntax & usage

The SUMIF function is used to conditionally sum values based on a **single criteria**. We discussed its syntax in detail in the [previous article](#), so let me give you just a quick summary now.

`SUMIF(range, criteria, [sum_range])`

**range** - the range of cells to be evaluated by your criteria, required.

**criteria** - the condition that must be met, required.

**sum\_range** - the cells to sum if the condition is met, optional.

As you see, the syntax of the Excel SUMIF function allows for one condition only. And still, we say that Excel SUMIF can be used to sum values with multiple criteria. How can that be? By adding the results of several SUMIF functions and by using SUMIF formulas with array criteria, as demonstrated in the examples that follow.

# Excel SUMIFS function - syntax & usage

You use SUMIFS in Excel to find a conditional sum of values based on **multiple criteria**. The SUMIFS function was introduced in Excel 2007, so you can use it in all modern versions of Excel 2016, 2013, 2010 and 2007.

Compared to SUMIF, the SUMIFS syntax is a little bit more complex:

`SUMIFS(sum_range, criteria_range1, criteria1, [criteria_range2, criteria2], ...)`

The first 3 arguments are mandatory, additional ranges and their associated criteria are optional.

**sum\_range** - one or more cells to sum, required. This can be a single cell, a range of cells or a named range. Only cells with numbers are summed; blank and text values are ignored.

**criteria\_range1** - the first range to be evaluated by the associated criteria, required.

**criteria1** - the first condition that must be met, required. You can supply the criteria in the form of a number, logical expression, cell reference, text or another Excel function. For example you can use criteria such as 10, ">=10", A1, "cherries" or TODAY().

`criteria_range2, criteria2, ...` - these are additional ranges and criteria associated with them, optional. You can use up to 127 range/criteria pairs in SUMIFS formulas.

**Important!** The SUMIFS function works with AND logic, meaning that each cell in the `sum_range` argument is summed only if all of the specified criteria are true for that cell.

And now, let's have a look at the Excel SUMIFS formula with two conditions. Suppose, you have a table listing the consignments of fruit from different suppliers. You have the fruit names in column A, suppliers' names in column B, and quantity in column C. What you want is to find out a sum of amounts relating to a given fruit and supplier, e.g. all apples supplied by Pete.

	A	B	C	D	E	F
1	Product	Supplier	Qty.		Product:	Apples
2	Cherries	John	200		Supplier:	Pete
3	Bananas	Mike	350		Qty.:	?
4	Apples	Pete	180			
5	Oranges	Mike	400			
6	Bananas	John	250			
7	Apples	Mike	120			
8	Cherries	John	330			
9	Apples	Pete	110			

When you're learning something new, it's always a good idea to start with simple things. So, to begin with, let's define all the arguments for our SUMIFS formula:

`sum_range` - C2:C9

`criteria_range1` - A2:A9




`criteria1` - "apples"

`criteria_range2` - B2:B9

`criteria2` - "Pete"

Now assemble the above parameters, and you will get the following SUMIFS formula:

`=SUMIFS(C2:C9, A2:A9, "apples", B2:B9, "Pete")`

F3	:	  	=SUMIFS(C2:C9, A2:A9, "apples", B2:B9, "Pete")				
	A	B	C	D	E	F	G
1	Product	Supplier	Qty.		Product:	Apples	
2	Cherries	John	200		Supplier:	Pete	
3	Bananas	Mike	350		Qty.:	290	
4	Apples	Pete	180				
5	Oranges	Mike	400				
6	Bananas	John	250				
7	Apples	Mike	120				
8	Cherries	John	330				
9	Apples	Pete	110				

criteria\_range1    criteria\_range2    sum\_range

Apples supplied by Pete:  
180+110=290

To refine the formula further, you can replace the text criteria "apples" and "Pete" with cell references. In this case, you won't have to change the formula to calculate the quantity of other fruit from a different supplier:

`=SUMIFS(C2:C9, A2:A9, F1, B2:B9, F2)`

## Using SUMIFS and SUMIF in Excel - things to remember

Since the aim of this tutorial is to cover all possible ways to sum values by several conditions, we will discuss formula examples with both functions - Excel SUMIFS and SUMIF with multiple criteria. To use them correctly, you need to clearly understand what these two functions have in common and in what way they are different.

While the common part is clear - similar destination and parameters - the differences are not so obvious, though very essential.

## 1. The order of arguments

In Excel SUMIF and SUMIFS functions, the order of arguments is different. In particular, sum\_range is the 1st parameter in SUMIFS, but it is 3<sup>rd</sup> in SUMIF.

At first sight, it may seem that Microsoft deliberately complicates the learning curve for its users. However, upon a closer look, you will see the reasoning behind it. The point is that sum\_range is optional in SUMIF. If you omit it, no problem, your SUMIF formula will sum values in the range (first parameter).

In SUMIFS, the sum\_range is very important and obligatory, and that is why it comes first. Probably Microsoft guys thought that after adding the 10<sup>th</sup> or 100<sup>th</sup> range / criteria pair, someone might forget to specify the range to sum : )

Wrapping up, if you are copying and editing these functions, make sure you put the parameters in the right order.

## 2. The size of sum\_range and criteria\_range arguments

In the SUMIF function, the sum\_range argument does not necessarily have to be of the same size as the range argument, as long as you have the top left cell right. In Excel SUMIFS, each criteria\_range must contain the same number of rows and columns as the sum\_range parameter.

For example, `=SUMIF(A2:A9, F1, C2:C18)` will return the correct result because Excel considers only the **upper leftmost** cell in the sum\_range argument (C2 in this example, which is correct), and then includes as many columns and rows as contained in the range argument.

The SUMIFS formula `=SUMIFS(C2:C9, A2:A9, "apples", B2:B10, "Pete")` will return the #VALUE! error because criteria\_range2 (B2:B10) does not match in size criteria\_range1 (A2:A9) and

sum\_range (C2:C9).

Alright, enough strategy (i.e. theory), let's get into the tactics (i.e. formula examples : )

## How to use SUMIFS in Excel - formula examples

A moment ago, we discussed a simple SUMIFS formula with two text criteria. In the same manner, you can use Excel SUMIFS with multiple criteria expressed by numbers, dates, logical expressions, and other Excel functions.

### Example 1. Excel SUMIFS with comparison operators

In our fruit suppliers table, suppose, you want to sum all deliveries by Mike with Qty. 200 or more. To do this, you use the comparison operator "greater than or equal to" ( $\geq$ ) in criteria2 and get the following SUMIFS formula:

`=SUMIFS(C2:C9,B2:B9,"Mike",C2:C9,">=200")`

F4

✕

✓

fx

=SUMIFS(C2:C9,B2:B9,"Mike",C2:C9,">=200")

	A	B	C	D	E	F	G
1	Product	Supplier	Qty.		Product:	Any	
2	Cherries	John	250		Supplier:	Mike	
3	Bananas	Mike	200		Qty.:	>200	
4	Apples	Pete	180		Total:	600	
5	Oranges	Mike	400				
6	Bananas	John	250				
7	Apples	Mike	120				
8	Cherries	John	330				
9	Apples	Pete	110				

Products delivered by Mike with Qty. 200 or more:

200+400=600

**Note.** Please pay attention that in Excel SUMIFS formulas, logical expressions with comparison



use the ampersand (&) to concatenate a string, for example "<="&TODAY().

In a similar fashion, you can use the Excel SUMIF function to sum values in a given date range. For example, the following SUMIFS formula adds the values in cells C2:C9 if a date in column B falls between 1-Oct-2014 and 31-Oct-2014, inclusive.

**=SUMIFS(C2:C9, B2:B9, ">=10/1/2014", B2:B9, "<=10/31/2014")**

The same result can be achieved by calculating the difference of two SUMIF functions, as demonstrated in this example - [How to use SUMIF to sum values in a given date range](#). However, Excel SUMIFS is much easier and more understandable, isn't it?

### Example 3. Excel SUMIFS with blank and non-blank cells

When analyzing reports and other data, you may often need to sum values corresponding either to empty or non-empty cells.

	Criteria	Description	Formula Example
Blank cells	"="	Sum values corresponding to blank cells that contain absolutely nothing - no formula, no zero length string.	<b>=SUMIFS(C2:C10, A2:A10, "=", B2:B10, "=")</b>  Sum values in cells C2:C10 if the corresponding cells in columns A and B are absolutely empty.
	""	Sum values corresponding to "visually" blank cells including those that contain empty strings returned by some other Excel function (for example, cells with a formula like ="").	<b>=SUMIFS(C2:C10, A2:A10, "", B2:B10, "")</b>  Sum values in cells C2:C10 with the same conditions as the above formula, but includes empty strings.
Non-blank	"<>"	Sum values corresponding to non-empty cells,	<b>=SUMIFS(C2:C10, A2:A10, "&lt;&gt;", B2:B10, "&lt;&gt;")</b>

cells		including zero length strings.	Sum values in cells C2:C10 if the corresponding cells in columns A and B are not empty, including cells with empty strings.
	SUM-SUMIF or SUM / LEN	Sum values corresponding to non-empty cells, not including zero length strings.	<b>=SUM(C2:C10) - SUMIFS(C2:C10, A2:A10, "", B2:B10, "")</b>  <b>=SUM((C2:C10) * (LEN(A2:A10)&gt;0)*(LEN(B2:B10)&gt;0))</b> Sum values in cells C2:C10 if the corresponding cells in columns A and B are not empty, cells with zero length strings are not included.

And now, let's see how you can use a SUMIFS formula with "blank" and "non-blank" criteria on real data. Suppose, you have an order date in column B, delivery date in column C and Qty. in column D. How do you find the total of products that have not been delivered yet? That is, you want to know the sum of values corresponding to non-empty cells in column B and empty cells in column C.

The solution is to use the SUMIFS formula with 2 criteria:

**=SUMIFS(D2:D10, B2:B10, "<>", C2:C10, "=")**



G2						=SUMIFS(D2:D10, B2:B10,"<>", C2:C10,"=")	
	A	B	C	D	E	F	G
1	Product	Order date	Delivery date	Qty.		Status:	Not delivered
2	Cherries	22-Sep	28-Sep	77		Qty.:	205
3	Bananas	4-Nov	13-Nov	52			
4	Apples	4-Dec		90			
5	Oranges	23-Oct	26-Oct	73			
6	Bananas	30-Oct		50			
7	Apples			58			
8	Cherries	8-Nov	17-Nov	90			
9	Apples	1-Dec		65			
10	Bananas	23-Dec	23-Dec	89			

Total of non-delivered products:  
90+50+65=205

## Using Excel SUMIF with multiple OR criteria

As noted in the beginning of this tutorial, the SUMIFS function is designed with AND logic. But what if you need to sum values with multiple OR criteria, i.e. when at least one of the conditions is met?

### Example 1. SUMIF + SUMIF

The simplest solution is to sum the results returned by several SUMIF functions. For example, the following formula demonstrates how to find the total of products delivered by Mike and John:

=SUMIF(C2:C9,"Mike",D2:D9) + SUMIF(C2:C9,"John",D2:D9)

=SUMIF(C2:C9,"Mike",D2:D9) + SUMIF(C2:C9,"John",D2:D9)

B	C	D	E	F	G	H
Product	Supplier	Qty.		Product:	Any	
Cherries	John	250		Supplier:	Mike	
Bananas	Mike	200		Supplier:	John	
Apples	Pete	180		Total:	1300	
Oranges	Mike	400				
Bananas	Pete	250				
Apples	Mike	120				
Cherries	John	330				
Apples	Pete	110				

Products delivered by Mike & John:  
250+200+400+120+330=1300

As you see, the first SUMIF function adds the quantities corresponding to "Mike", the other SUMIF function returns the amounts relating to "John" and then you add these 2 numbers.

## Example 2. SUM & SUMIF with an array argument

The above solution is very simple and may get the job done quickly when there are only a couple of criteria. But a SUMIF + SUMIF formula may grow up enormously if you want to sum values with multiple OR conditions. In this case, a better approach is using an **array criteria** argument in the SUMIF function. Let's examine this approach now.

You can start by listing all of your conditions separated by commas and then enclose the resulting comma-separated list in {curly brackets}, which is technically called an array.

In the previous example, if you want to sum the products delivered by John, Mike and Pete, your array criteria will look like {"John","Mike","Pete"}. And the complete SUMIF function is **SUMIF (C2 : C9 , {"John" , "Mike" , "Pete"} , D2 : D9) .**



### Example 3. SUMPRODUCT & SUMIF

In case, your preferred way is to list the criteria in some cells rather than specify them directly in the formula, you can use SUMIF in conjunction with the SUMPRODUCT function that multiplies components in the given arrays, and returns the sum of those products.

`=SUMPRODUCT(SUMIF(C2:C9, G2:G4, D2:D9))`

Where G2:G4 are the cells containing your criteria, the suppliers' names in our case, as illustrated in the screenshot below.

But of course, nothing prevents you from listing the values in an array criteria of your SUMIF function if you want to:

`=SUMPRODUCT(SUMIF(C2:C9, {"Mike", "John", "Pete"}, D2:D9))`

The result returned by both formulas will be identical to what you see in the screenshot:

The screenshot shows an Excel spreadsheet with a formula bar at the top containing `=SUMPRODUCT(SUMIF(C2:C9, G2:G4, D2:D9))`. Below the formula bar is a table with columns B, C, D, E, F, G, H, and I. The table has a header row with green backgrounds for columns B, C, and D. The data rows show product quantities for different suppliers. An orange arrow points from the formula bar to cell G2, and another orange arrow points from the word 'Criteria' to the range G2:G4. The total value 1060 is shown in cell G5.

B	C	D	E	F	G	H	I
Product	Supplier	Qty.		Product:	Any		
Cherries	John	250		Supplier:	Mike		
Bananas	Sally	200		Supplier:	John		
Apples	Pete	180		Supplier:	Pete		
Oranges	Mike	400		Total:	1060		
Bananas	Sally	250					
Apples	Mike	120					
Cherries	Sally	330					
Apples	Pete	110					

## Excel SUMIFS with multiple OR criteria

If you want to conditionally sum values in Excel not simply with multiple OR conditions, but with several sets of conditions, you will have to use SUMIFS instead of SUMIF. The formulas will be very similar to what we've just discussed.

As usual, an example might help to illustrate the point better. In our table of fruit suppliers, let's add the Delivery Date (column E) and find the total quantity delivered by Mike, John and Pete in October.

### Example 1. SUMIFS + SUMIFS

The formula produced by this approach includes a lot of repetition and looks cumbersome, but it is easy to understand and, most importantly, it works : )

```
=SUMIFS(D2:D9,C2:C9, "Mike", E2:E9,">=10/1/2014", E2:E9, "<=10/31/2014") +  
SUMIFS(D2:D9, C2:C9, "John", E2:E9, ">=10/1/2014", E2:E9, "<=10/31/2014") +  
SUMIFS(D2:D9, C2:C9, "Pete", E2:E9, ">=10/1/2014", E2:E9, "<=10/31/2014")
```

As you see, you write a separate SUMIFS function for each of the suppliers and include two conditions - equal to or greater than Oct-1 (">=10/1/2014",) and less than or equal to Oct 31 ("<=10/31/2014"), and then you sum the results.

```
=SUMIFS(D2:D9,C2:C9, "Mike", E2:E9,">=10/1/2014", E2:E9, "<=10/31/2014") +  
SUMIFS(D2:D9, C2:C9, "John", E2:E9, ">=10/1/2014", E2:E9, "<=10/31/2014") +  
SUMIFS(D2:D9, C2:C9, "Pete", E2:E9, ">=10/1/2014",E2:E9, "<=10/31/2014")
```

B	C	D	E	F	G	H
Product	Supplier	Qty.	Delivery Date		Month:	October
Cherries	John	250	29-Oct		Supplier 1:	Mike
Bananas	Sally	200	12-Nov		Supplier 2:	John
Apples	Pete	180	29-Oct		Supplier 3:	Pete
Oranges	Mike	400	5-Nov		Total:	660
Bananas	Sally	250	4-Oct			
Apples	Mike	120	5-Oct			
Cherries	Sally	330	24-Sep			
Apples	Pete	110	30-Oct			

Products delivered by Mike,  
John and Pete in October:  
  
250+180+120+110=660

## Example 2. SUM & SUMIFS with an array argument

I've tried to explain the essence of this approach in the [SUMIF example](#), so now we can simply copy that formula, change the order of arguments (as you remember it is different in SUMIF and SUMIFS) and add additional criteria. The resulting formula is more compact than SUMIFS + SUMIFS:

```
=SUM(SUMIFS(D2:D9,C2:C9, {"Mike", "John", "Pete"}, E2:E9,">=10/1/2014",  
E2:E9, "<=10/31/2014"))
```

The result returned by this formula is exactly the same as you see in the screenshot above.

## Example 3. SUMPRODUCT & SUMIFS

As you remember, the SUMPRODUCT approach differs from the previous two in the way that you enter



turning them into +1 and 0, which the SUMPRODUCT function can understand.

I hope the above explanation makes sense. And even if it doesn't, just remember this rule of thumb - use the double unary operator (--) when you are using comparison operators in your SUMPRODUCT formulas.

## Using Excel SUM in array formulas

As you remember, Microsoft implemented the SUMIFS function in Excel 2007. If someone still uses Excel 2003, 2000 or earlier, you will have to use a SUM array formula to add values with multiple AND criteria. Naturally, this approach works in modern versions of Excel 2013 - 2007 too, and can be deemed an old-fashioned counterpart of the SUMIFS function.

In the [SUMIF formulas](#) discussed above, you have already used array arguments, but an array formula is something different.

### Example 1. Sum with multiple AND criteria in Excel 2003 and earlier

Let's get back to the [very first example](#) where we found out a sum of amounts relating to a given fruit and supplier:

	A	B	C	D	E	F
1	Product	Supplier	Qty.		Product:	Apples
2	Cherries	John	200		Supplier:	Pete
3	Bananas	Mike	350		Qty.:	?
4	Apples	Pete	180			
5	Oranges	Mike	400			
6	Bananas	John	250			
7	Apples	Mike	120			
8	Cherries	John	330			
9	Apples	Pete	110			



As you already know, this task is easily accomplished using an ordinary SUMIFS formula:

`=SUMIFS(C2:C9, A2:A9, "apples", B2:B9, "Pete")`

And now, let's see how the same task can be fulfilled in early "SUMIFS-free" versions of Excel. First off, you write down all the conditions that should be met in the form of range="condition". In this example, we have two range/condition pairs:

Condition 1: A2:A9="apples"

Condition 2: B2:B9="Pete"

Then, you write a SUM formulas that "multiplies" all of your range/condition pairs, each enclosed in brackets. The last multiplier is the sum range, C2:C9 in our case:

`=SUM((A2:A9="apples") * ( B2:B9="Pete") * ( C2:C9))`

As illustrated in the screenshot below, the formula perfectly works in the latest Excel 2013 version.

	A	B	C	D	E	F	G
1	Product	Supplier	Qty.		Product:	Apples	
2	Cherries	John	200		Supplier:	Pete	
3	Bananas	Mike	350		Qty.:	290	
4	Apples	Pete	180				
5	Oranges	Mike	400				
6	Bananas	John	250				
7	Apples	Mike	120				
8	Cherries	John	330				
9	Apples	Pete	110				

Using SUM array formula to sum with multiple criteria.

**Note.** When entering any array formula, you must press Ctrl + Shift + Enter. Once you do this, your formula gets enclosed in {curly braces}, which is a visual indication that an array formula is entered

correctly. If you try typing the braces manually, your formula will be converted to a text string, and it won't work.

## Example 2. SUM array formulas in modern Excel versions

Even in modern versions of Excel 2016, 2013, 2010 or 2007, the power of the SUM function should not be underestimated. The SUM array formula is not simply gymnastics of the mind, but has a practical value, as demonstrated in the following example.

Suppose, you have two columns, B and C, and you need to count how many times column C is greater than column B, when a value in column C is greater or equal to 10. An immediate solution that comes to mind is using the SUM array formula:

`=SUM((C1:C10>=10) * (C1:C10>B1:B10))`

{=SUM((C1:C10>=10)*(C1:C10>B1:B10))}				
B	C	D	E	F
6	24		C>=10 and C>B	3
44	45			
20	12			
18	7			
12	12			
36	5			
47	13			
4	16			
11	8			
45	23			

How many times values in column C are greater than in column B when C >= 10.

Don't see any practical application to the above formula? Think about it in another way : )

Suppose, you have the orders list like shown in the screenshot below and you want to know how many

products have not been delivered in full by a given date. Translated into Excel's language, we have the following conditions:

Condition 1: A value in column B (Ordered items) is greater than 0

Condition 2: A value in column C (Delivered) is less than in column B

Condition 3: A date in column D (Due date) is less than 11/1/2014.

Putting the three range/condition pairs together, you get the following formula:

`=SUM( (B2:B10>=0) * (B2:B10>C2:C10) * (D2:D10<G2) )`

G1

:

✕

✓

fx

{=SUM((B2:B10>=0)\*(B2:B10>C2:C10)\*(D2:D10<G2))}

	A	B	C	D	E	F	G
1	Product	Ordered	Delivered	Due date		Not delivered in full:	3
2	Cherries	44	44	29-Oct		Due date:	1-Nov
3	Bananas	20	12	12-Oct			
4	Apples	18	7	29-Oct			
5	Oranges	12	12	5-Oct			
6	Lemons	0	0			<div>The formula counts how many products had not been delivered in full by Nov-1.</div>	
7	Kiwis	47	13	3-Dec			
8	Mangos	0	0				
9	Peaches	11	8	30-Oct			
10	Grapefruit	45	45	4-Oct			